

Identities involving Spherical Bessel Functions

While integrating the internal electromagnetic fields inside a sphere, several integrals of products of spherical Bessel functions arise. These integrals are in turn proportional to the transition matrix elements between the internal and the scattered fields of the sphere. To the best of my knowledge, these relationships are not present in the literature. This may be due to the fact that the far field in Mie theory can simply be obtained by integrating the scattered field over a unit sphere via the following expression

$$\mathbf{F}(\hat{\mathbf{k}}) = \frac{ika^2}{4\pi} \int \hat{\mathbf{k}} \times \left(\hat{\mathbf{s}} \times \mathbf{E}_{\text{sca}}(\mathbf{s}) + \eta_0 \hat{\mathbf{k}} \times \hat{\mathbf{s}} \times \mathbf{H}_{\text{sca}}(\mathbf{s}) \right) d\hat{\mathbf{s}} \quad (1)$$

This expression can be evaluated on any spherical surface of radius “a” that encloses the actual sphere. However, in the application of Mie theory relevant to my work in this project, I need to evaluate \mathbf{F} via the following expression over the volume of the sphere,

$$\mathbf{F}(\hat{\mathbf{k}}) = \frac{k^2}{4\pi} (n^2 - 1) \int_{\text{sphere}} (\mathbf{I} - \hat{\mathbf{k}} \hat{\mathbf{k}}) \cdot \mathbf{E}_{\text{sph}}(\mathbf{s}) d\mathbf{s} \quad (2)$$

This expression leads to three integrals that must be evaluated in a general form to confirm that the internal field volume integral in the above expression yields exactly the same result as the surface integral. Of course, this is ensured by the surface equivalence theorem. However, my purpose here is to validate the numerical implementation of both ways of calculating the scattering matrix. For this, I find it reassuring to first learn how the expressions that follow from (1) will follow from (2). The details are outlined in my notes. Here I only compute the two types of integrals arising in the analysis:

$$A_l(n, z) = \int_0^z x^2 j_l(x) j_l(nx) dx$$

$$B_l(n, z) = (n^2 - 1) \int_0^z [\psi'(x) \psi'(nx) + l(l+1) j_l(x) j_l(nx)] dx$$

Definitions of elementary functions

First I define the Riccati functions and their derivatives so that the expressions above can be formed easily. I use the symbol ψ for the function $x j_l(x)$ and the symbol $\xi(x) = x h^{(1)}(x)$ where $h^{(1)}$ is the spherical Hankel function of the first kind.

```

In[1]:=  $\psi[l_, z_] := z \text{SphericalBesselJ}[l, z]$ 
 $\xi[l_, z_] := z \text{SphericalHankelH1}[l, z]$ 
 $D\psi[l_, z_] := \text{Evaluate}[D[\psi[l, z], z]]$ 
 $D\xi[l_, z_] := \text{Evaluate}[D[\xi[l, z], z]]$ 

 $\alpha_1[l_, n_, z_] := \psi[l, z] \times D\psi[l, n z] - 1/n D\psi[l, z] \times \psi[l, n z]$ 
 $\alpha_2[l_, n_, z_] := \psi[l, z] \times D\psi[l, n z] - n D\psi[l, z] \times \psi[l, n z]$ 
 $\beta_1[l_, n_, z_] := \xi[l, z] \times D\psi[l, n z] - 1/n D\xi[l, z] \times \psi[l, n z]$ 
 $\beta_2[l_, n_, z_] := \xi[l, z] \times D\psi[l, n z] - n D\xi[l, z] \times \psi[l, n z]$ 

 $T_{\text{scaM}}[l_, n_, z_] := -\frac{\alpha_1[l, n, z]}{\beta_1[l, n, z]}$ 
 $T_{\text{scaN}}[l_, n_, z_] := -\frac{\alpha_2[l, n, z]}{\beta_2[l, n, z]}$ 

```

Integral $A_l(n, z)$

The first integral is over spherical Bessel functions. It has a closed form expression. The denominator going to zero as $n \rightarrow 1$ is a removable singularity as is confirmed below using the limit function.

```

In[11]:= A = Simplify[(n^2 - 1) Integrate[SphericalBesselJ[l, n x] SphericalBesselJ[l, x] x^2,
      x, Assumptions -> {l > 1, l ∈ Integers, x ∈ Reals}]];
TraditionalForm[
  A]

```

Out[12]//TraditionalForm=

$$\frac{\pi x^{3/2} \left(J_{l-\frac{1}{2}}(x) J_{l+\frac{1}{2}}(nx) - n J_{l+\frac{1}{2}}(x) J_{l-\frac{1}{2}}(nx) \right)}{2 \sqrt{nx}}$$

The above expression is equivalent to the following, after I substitute the spherical Bessel function for the Bessel functions of half-integer order above:

```

In[13]:= RHS1[l_, n_, z_] := n^2 (SphericalBesselJ[l - 1, n z] SphericalBesselJ[l, z] -
      1 / n SphericalBesselJ[l - 1, z] SphericalBesselJ[l, n z])

```

Verify that the result above is equal to $\alpha_1(n, z)$

```

In[14]:= FullSimplify[alpha1[l, n, z] - RHS1[l, n, z]]

```

Out[14]= 0

Integral $B_l(n, z)$

Taylor Series Expansion at $z = 0$

This integral has no direct analytical expression available. I first check that the expressions

```

In[15]:= B[l_, n_, z_] := -(n^2 - 1)
        (Dψ[l, n z] × Dψ[l, z] + l (l + 1) SphericalBesselJ[l, n z] SphericalBesselJ[l, z])

In[16]:= lval = 1;
maxterm = 3;
temp1 = Series[B[lval, n, z], {z, 0, maxterm}, Assumptions → {z ∈ Reals, z > 0}];
temp1 = FullSimplify[Integrate[Normal[temp1], z]];
temp2 =
    Series[alpha2[lval, n, z], {z, 0, maxterm}, Assumptions → {z ∈ Reals, z > 0}];
temp2 = FullSimplify[Normal[temp2]];

Print["Taylor expansion:\nBl(l,n,z) = ", TraditionalForm[temp1]]
Print["α2l(l,n,z) = ", TraditionalForm[temp2]]
FullSimplify[temp1 - temp2]

Taylor expansion:

$$B_l(l, n, z) = -\frac{2}{9} n (n^2 - 1) z^3$$


$$\alpha_{2l}(l, n, z) = -\frac{2}{9} n (n^2 - 1) z^3$$


Out[24]= 0

```

Numerical Integration of $B_l(n, z)$

Using numerical integration, I verify below that the integral $B_l(n, z) = \alpha_{2l}(n, z)$ for any arbitrary value of n, z, l . One can change the values of each of these parameters and find that the difference between numerically evaluated $B_l(n, z)$ is equal to $\alpha_{2l}(n, z)$ within the numerical precision.

```

In[25]:= zval = 3;
nval = 1.5 + I;
temp1 = Table[NIntegrate[B[l, nval, z], {z, 0, zval}], {l, 1, 5}];
temp2 = Table[alpha2[l, nval, zval], {l, 1, 5}];
Print["Bl(n,z)-α2l(n,z) = ", temp1 - temp2]

Bl(n,z)-α2l(n,z) = {-7.99361 × 10-15 - 8.88178 × 10-15 i,
-1.15463 × 10-14 - 7.99361 × 10-15 i, -2.10942 × 10-15 - 1.06581 × 10-14 i,
2.88658 × 10-15 - 2.77556 × 10-15 i, 7.77156 × 10-16 + 5.55112 × 10-17 i}

```